unittest
{
	assert( [ "4 3", "3553" ].parse.expand.solve == 2 );		// 追加ケース
	assert( [ "4 2", "1221" ].parse.expand.solve == 5 );		// 追加ケース
	
	assert( [ "4 3", "3543" ].parse.expand.solve == 6 );
	assert( [ "4 2", "2020" ].parse.expand.solve == 10 );
	assert( [ "20 11", "33883322005544116655" ].parse.expand.solve == 68 );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	auto np = input.front.split.to!( long[] );
	input.popFront;
	auto ss = input.front.map!( ( a ) => ( a - '0' ).to!long ).array;
	return tuple( np[ 1 ], ss );
}

auto solve( long p, long[] ss )
{
	auto ans = 0L;
	if( 10 % p == 0 )
	{
		// 10 が p で割り切れる場合、10の位以上は絶対に p で割り切れるので、1の位だけ見て判定する
		foreach( i, s; ss )
		{
			if( s % p == 0 ) ans += i + 1;
		}
	}
	else
	{
		// 10 と p が互いに素である場合、1桁ずつ余りを取って、余りごとにカウントする
		auto ts = new long[ p ];
		auto m = 0L;
		foreach( i, s; ss )
		{
			m = ( m * 10 + s ) % p;
			ts[ m ]++;
		}
		m = 0L;
		foreach( i, s; ss )
		{
			ans += ts[ m ];
			m = ( m * 10 + s ) % p;
			ts[ m ]--;
		}
	}
	return ans;
}
